

ANEXO 2

Código fonte do SAS© para extração e tratamento da
base de dados do SIB de beneficiários

```

*****
/* Calcula a média mensal de 12 meses dos beneficiários para o CMH */
* Rateio dos beneficiários em planos sem indicação do tipo de financiamento
*****;

*****!!!!!!!!!!!!!!!!!!!!!! FILTRO !!!!!!!!!!!!!!!!!!!!!!!*****;
* Filtra a base de beneficiários relevante do arquivo RPS_SIB_FE;
* exclui operadoras odontológicas e administradoras;
* exclui planos exclusivamente odontológicos;
* Identifica os grupos a serem rateados: Planos sem identificação de pré/pós;;
PROC SQL;
CREATE TABLE a1.SIB AS
SELECT t1.CD_OPERADORA,
       case
         when t1.ds_classificacao = "Autogestão" then t1.ds_classificacao
         else "Outros"
       end as modalidade,
CASE
  WHEN de_tipo_contratacao = "Individual ou familiar" and tp_vigencia_plano = "A" THEN 'ind_a'
  WHEN de_tipo_contratacao = "Individual ou familiar" and tp_vigencia_plano = "P" THEN 'ind_p'
  WHEN de_tipo_contratacao = "Individual ou familiar" THEN 'rat_ind'

  WHEN de_tipo_contratacao = "Coletivo empresarial" and tp_vigencia_plano = "A" then "emp_a"
  WHEN de_tipo_contratacao = "Coletivo empresarial" and de_tipo_modalidade_fim = "Pre-estabelecido" then "emp_p_pre"
  WHEN de_tipo_contratacao = "Coletivo empresarial" and de_tipo_modalidade_fim = "Pós-estabelecido" then "emp_p_pos"
  WHEN de_tipo_contratacao = "Coletivo empresarial" then "rat_emp_p"

  WHEN de_tipo_contratacao = "Coletivo por adesão" and tp_vigencia_plano = "A" then "ade_a"
  WHEN de_tipo_contratacao = "Coletivo por adesão" and de_tipo_modalidade_fim = "Pre-estabelecido" then "ade_p_pre"
  WHEN de_tipo_contratacao = "Coletivo por adesão" and de_tipo_modalidade_fim = "Pós-estabelecido" then "ade_p_pos"
  WHEN de_tipo_contratacao = "Coletivo por adesão" then "rat_ade_p"

ELSE ""
END AS tipo format=$9.,
       t1.MES,

(SUM(t1.BENEF)) FORMAT=COMMAX20. AS BENEF

FROM a1.RPS_SIB_FE t1
  where id_tipo_segmentacao ne 4
         and t1.ds_classificacao not in ('Cooperativa odontológica','Odontologia de Grupo',
                                         'Administradora','Administradora de Benefícios')
         and t1.mes > 201300

GROUP BY t1.CD_OPERADORA,
         calculated modalidade,
         calculated tipo,
         t1.MES
;
QUIT;

*****!!!!!!!!!!!!!!!!!!!!!! FILTRO !!!!!!!!!!!!!!!!!!!!!!!*****;
* Exclui operadoras em início de operação;;
PROC SQL;
CREATE TABLE BENEF0 AS
SELECT t1.*
FROM a1.SIB t1 left join a1.registro t2 on t1.cd_operadora=t2.cd_operadora
       where t1.mes > (t2.dt_inicio)
;
QUIT;

*****!!!!!!!!!!!!!!!!!!!!!! FILTRO !!!!!!!!!!!!!!!!!!!!!!!*****;
* Exclui operadoras com registro cancelado;;
PROC SQL;
CREATE TABLE BENEF1 AS
SELECT t1.*
FROM benef0 t1 left join a1.CANCELAMENTOS t2 on t1.cd_operadora=t2.cd_operadora
       where t2.cd_operadora is missing or t1.mes < (t2.dt_cancelamento-2)
;
QUIT;

*****!!!!!!!!!!!!!!!!!!!!!! FILTRO !!!!!!!!!!!!!!!!!!!!!!!*****;
* exclui planos não adaptados
  exclui planos do SIB não identificados no RPS;;
data benef2 x1 x2;
  set benef1;

  if tipo = "" then output x1;
  else if benef < 1 then output x2;
  else output benef2;

run;

*****;
*****;
*****;
* RATEIOS;
* nota: podem haver planos que não têm classificação de tipo de modalidade de financiamento
  critério - rateio proporcional pela distribuição dos planos com informação;;

```

```

* joga tipo de plano para colunas;;
proc sort data=benef2; by cd_operadora modalidade mes; run;
proc transpose data=benef2 out=benef3 (drop=_name_) suffix=_0;
  by cd_operadora modalidade mes;
  id tipo;
  var BENEFF;
run;
proc stdize out=benef3 reponly missing=0; run; * transforma vazios em zeros;

* calcula proporção de todas as operadoras para rateio dos casos em que o valor a ser rateado
  seja o único valor informado pela operadora;;
* Prepara tabelas de % por operadora e geral do mercado para rateio;;
PROC SQL;
* a) tabela % por operadora;
create table pct_tt_ops0 as
select CD_OPERADORA, modalidade, mes,
  SUM(ade_p_pre_0) AS ade_p_pre,
  sum(ade_p_pos_0) AS ade_p_pos,
  sum(emp_p_pre_0) AS emp_p_pre,
  sum(emp_p_pos_0) AS emp_p_pos,
  sum(ade_p_pos_0, ade_p_pre_0) AS ade_p_tt,
  sum(emp_p_pos_0, emp_p_pre_0) AS emp_p_tt

from benef3
group by CD_OPERADORA, modalidade, mes;

create table pct_tt_ops as
select CD_OPERADORA, modalidade, mes,
  ade_p_pre/ ade_p_tt as pct_ade_p_pre_ops format=percent20.1,
  ade_p_pos/ ade_p_tt as pct_ade_p_pos_ops format=percent20.1,

  emp_p_pre/ emp_p_tt as pct_emp_p_pre_ops format=percent20.1,
  emp_p_pos/ emp_p_tt as pct_emp_p_pos_ops format=percent20.1

from pct_tt_ops0
  order by modalidade, mes;

* b) tabela % do mercado;
create table pct_tt0 as
select modalidade, mes,
  SUM(ade_p_pre) AS ade_p_pre,
  sum(ade_p_pos) AS ade_p_pos,
  sum(emp_p_pre) AS emp_p_pre,
  sum(emp_p_pos) AS emp_p_pos,
  sum(ade_p_tt) AS ade_p_tt,
  sum(emp_p_tt) AS emp_p_tt
from pct_tt_ops0
group by modalidade, mes;

create table pct_tt as
select modalidade, mes,
  ade_p_pre/ ade_p_tt as pct_ade_p_pre format=percent20.1,
  ade_p_pos/ ade_p_tt as pct_ade_p_pos format=percent20.1,
  emp_p_pre/ emp_p_tt as pct_emp_p_pre format=percent20.1,
  emp_p_pos/ emp_p_tt as pct_emp_p_pos format=percent20.1
from pct_tt0
  order by modalidade, mes;

quit;

data pct_tt;
merge pct_tt_ops pct_tt;
  by modalidade mes;
run;

* Executa o Rateio;;
proc sort data=pct_tt; by cd_operadora modalidade mes; run;
proc sort data=benef3 (rename=(ind_a_0=ind_a ind_p_0=ind_p emp_a_0=emp_a ade_a_0=ade_a));
  by cd_operadora modalidade mes; run;

data benef4;
merge benef3 pct_tt;
by cd_operadora modalidade mes;
format ade_p_pos ade_p_pre emp_p_pos emp_p_pre commax20.;

* Rateio de coletivo por adesão;
* se não tiver o que ratear, assume o valor como está;
if rat_ade_p_0 = 0 then do;
  ade_p_pos=ade_p_pos_0;
  ade_p_pre=ade_p_pre_0;
end;

* se tiver o que ratear...;
else do;
* rateia usando o % de todas as operadoras, se não tiver % da própria operadora;
  if sum(ade_p_pos_0, ade_p_pre_0)=0 then do;
    ade_p_pos = pct_ade_p_pos * rat_ade_p_0;
    ade_p_pre = pct_ade_p_pre * rat_ade_p_0;
  end;
* ou rateia usando o % da própria operadora;
  else do;
    ade_p_pos = pct_ade_p_pos_ops * rat_ade_p_0 + ade_p_pos_0;
    ade_p_pre = pct_ade_p_pre_ops * rat_ade_p_0 + ade_p_pre_0;
  end;
end;

```

```

end;

* Rateio de coletivo empresarial;
* se não tiver o que ratear, assume o valor como está;
if rat_emp_p_0 = 0 then do;
    emp_p_pos=emp_p_pos_0;
    emp_p_pre=emp_p_pre_0;
end;

* se tiver o que ratear...;
else do;
* rateia usando o % de todas as operadoras, se não tiver % da própria operadora;
    if sum(emp_p_pos_0, emp_p_pre_0)=0 then do;
        emp_p_pos = pct_emp_p_pos * rat_emp_p_0;
        emp_p_pre = pct_emp_p_pre * rat_emp_p_0;
    end;
* ou rateia usando o % da própria operadora;
    else do;
        emp_p_pos = pct_emp_p_pos_ops * rat_emp_p_0 + emp_p_pos_0;
        emp_p_pre = pct_emp_p_pre_ops * rat_emp_p_0 + emp_p_pre_0;
    end;
end;

keep cd_operadora mes ade_a ade_p_pos ade_p_pre emp_a emp_p_pos emp_p_pre ind_a ind_p;
run;

* volta tipo de plano para linhas;;
proc sort data=benef4; by cd_operadora mes; run;
proc transpose data=benef4 out=benef5 (rename=(coll=BENEF _name_=tipo));
by cd_operadora mes;
run;

* Exclui observações com menos de 1 beneficiário;;
data al.SIB_Tratado;
set benef5;

where benef ne . and benef >= 1;
chave = compress(cd_operadora)||" "||strip(tipo);

keep cd_operadora tipo chave mes benef;
run;
* FIM DO RATEIO;

* CALCULA A MÉDIA MENSAL DOS ÚLTIMOS 12 MESES DE CADA PERÍODO;
* calcula o total de beneficiários dos últimos 12 meses ;;
proc sort data=al.SIB_Tratado; by chave mes; run;
data benef7;
* varrer as datas de cada chave;
do i = 1 by 1 until (last.chave);
set al.SIB_Tratado;
by chave;
format benef01 benef02 benef03 benef04 benef05 benef06 benef07
    benef08 benef09 benef10 benef11 benef12 benefM commax20.;

if first.chave then do;
    benef12=0; benef11=0; benef10=0; benef09=0; benef08=0; benef07=0;
    benef06=0; benef05=0; benef04=0; benef03=0; benef02=0;
    benef01=benef;
    benefM=sum(benef);

    mes0=mes;
    loop = 1;
    output;
end;

else if (mes-mes0=1) or
((substr(put(mes,6.),5,2)='01') and (mes-mes0 = 89)) then do;
    benef12=benef11;
    benef11=benef10;
    benef10=benef09;
    benef09=benef08;
    benef08=benef07;
    benef07=benef06;
    benef06=benef05;
    benef05=benef04;
    benef04=benef03;
    benef03=benef02;
    benef02=benef01;
    benef01=benef;
    benefM=sum(benef01,benef02,benef03,benef04,benef05,benef06,benef07,benef08,benef09,benef10,benef11,benef12);

    mes0=mes;
    loop = loop+1;
    output;
end;

else do;
    benef12=0; benef11=0; benef10=0; benef09=0; benef08=0;
    benef07=0; benef06=0; benef05=0; benef04=0; benef03=0; benef02=0;
    benef01=benef;
    benefM=sum(benef);

    mes0=mes;
    loop = 1;
    output;
end;
end;

```

```

end;
run;

*****!!!!!!!!!!!!!!!!!!!!!! FILTRO !!!!!!!!!!!!!!!!!!!!!!!*****;
* Só considera casos em que a operadora apresentou 12 meses ininterruptos de beneficiários no SIB > 1;
data al.benefM;
  set benef7;

  where loop >= 12;

  if benefM <= 1 then benefM = 0;
  trimestre = mes;
  benefM = benefM / 12;

  if substr(put(trimestre,6.),5,2) in ('03','06','09','12');

  if tipo in ("ind_a", "ade_a", "emp_a") then TP_vigencia_plano = "A";
  else TP_vigencia_plano = "P";

  if tipo in ("ind_a", "ind_p") then de_tipo_contratacao = "Individual ou familiar";
  else if tipo in ("ade_a","ade_p_pre","ade_p_pos") then de_tipo_contratacao = "Coletivo por adesão";
  else if tipo in ("emp_a","emp_p_pre","emp_p_pos") then de_tipo_contratacao = "Coletivo empresarial";
  else de_tipo_contratacao = "ERRO";

  If tipo in ("ind_a","ind_p","ade_p_pre","emp_p_pre") then de_tipo_modalidade_finm = "Pre-estabelecido";
  else if tipo in ("ade_p_pos","emp_p_pos") then de_tipo_modalidade_finm = "Pós-estabelecido";
  else if tipo in ("ade_a","emp_a") then de_tipo_modalidade_finm = "Plano antigo";
  else de_tipo_modalidade_finm = "ERRO";

  keep cd_operadora tipo trimestre benefM
  tp_vigencia_plano de_tipo_contratacao de_tipo_modalidade_finm;
run;

```